
Abstract

This short text introduces the essence of Coq as a formal system. It does not attempt (in its current
form) to introduce the features necessary to use Coq for real work. Prerequisite knowledge: basic
programming.

1 Motivation

Our research goal is to prove certain properties of certain kinds of programs, i.e. to establish beyond a
reasonable doubt that these properties hold for these programs. But when have we established something
beyond a reasonable doubt?

A very rigorous way to address this issue is to use a mechanised proof system: if we can express the
statements that we wish to prove, as well as the proofs themselves, as strings that we can input into a
computer, then we can write a program that checks that a given string is a proof of a given statement.

Such programs have in fact been written; they are called proof assistants. One of the most popular proof
assistants is Coq1. In order to introduce the notion of formal logic and formal proof, in the remainder of
this text we will give an introduction to the Coq proof assistant.

2 Data values and functions

In a first approximation, Coq is just a programming language. (We will see soon how this relates to formal
proof.) More specifically, it is a functional programming language, not unlike Haskell or ML. In Coq, there
are two kinds of values: data values (such as booleans, integers, and strings) and functions. Each value has
a type; for example, the type of true is bool. There are two kinds of types: datatypes (also called inductive
types for reasons that will become clear later) and function types. The data values are the values of the
datatypes and the functions are the values of the function types.

We illustrate these concepts using the following simple example:

Inductive bool := true | false.

Definition negate(b: bool): bool :=
match b with
true => false

| false => true
end.

This Coq input file defines the datatype bool and the constant negate, which is bound to a function
value that takes a bool argument and returns its negation. A datatype definition starts with the keyword
Inductive and lists the constructors of the datatype. The constructors of type bool are true and false.
Each constructor of datatype bool yields a value of type bool.

Note: to make it clearer that constant negate is bound to a value of a function type, we can rewrite the
definition to the following equivalent form:

Definition negate: bool -> bool :=
fun (b: bool) =>
match b with
true => false

| false => true
end.

The type of negate is bool -> bool, the type of functions that take a bool and return a bool. The
lambda expression fun (x: T) => E denotes the function that takes an argument x of type T and returns
the value of expression E (that can use x). In the example, the function uses a match expression to return a
different result depending on the constructor used to construct the bool value b.

To apply a function to an argument, simply write them next to each other, separated by a space:

Compute negate true. (* Output: = false: bool *)

This snippet uses the Compute command to evaluate the expression negate true, which applies function
negate to argument true. If you run this command in Coq, Coq will report that the result is the value
false, of type bool.

1http://coq.inria.fr/

3 Parameterized constructors

Datatype constructors can be parameterized. Such constructors yield a different value for each argument
sequence. For example, consider the type byte:

Inductive byte := Byte(b7 b6 b5 b4 b3 b2 b1 b0: bool).

Definition negate_byte(b: byte): byte :=
match b with
Byte b7 b6 b5 b4 b3 b2 b1 b0 =>
Byte

(negate b7) (negate b6) (negate b5) (negate b4)
(negate b3) (negate b2) (negate b1) (negate b0)

end.

The datatype byte has 256 values, all yielded by applying the single constructor Byte to different se-
quences of booleans. Function negate_byte uses a match expression to bind the arguments used to construct
the byte value b to the variables b7, b6, etc.

4 Recursive datatypes

Datatype constructors can take values of the datatype being defined as arguments; that is, a datatype
definition can refer to itself in its definition. This can be used to define the type of strings of bytes, as
follows:

Inductive string := EmptyString | NonemptyString(head: byte)(tail: string).

Definition h: byte := Byte false true true false true false false false. (* hex 68 *)
Definition i: byte := Byte false true true false true false false true. (* hex 69 *)

Definition hi: string := NonemptyString h (NonemptyString i EmptyString).

Notice that datatype string appears in its own definition as the type of the second parameter of con-
structor NonemptyString. As a result, a given value of type string can be used to construct a larger value
of type string. This is illustrated by the constant hi, which is bound to the ASCII encoding of the two
letters “hi”. The string “hi” is constructed from the string “i”, which is constructed from the empty string.

Let us define a function that concatenates two strings:

Definition concat(s1 s2: string): string :=
match s1 with

EmptyString => s2
| NonemptyString b s => NonemptyString b (concat s s2)
end.

Function concat inspects the first string: if it is empty, the result is simply the second string. Otherwise,
the result is constructed from the first byte (the head) of the first string and the concatenation of the rest
(the tail) of the first string and the second string.

Unfortunately, Coq does not allow the recursive call of concat. To ensure that constants are well-defined,
a constant cannot appear in its own definition. This rules out the following pathological definition:

Definition b: bool := negate b.

But then, how do we define the string concatenation function? The answer is that Coq provides a fix-
point operator fix f(x: T) {struct x}: U := E which denotes the function that takes an argument x
of type T and returns the value of the expression E of type U. The difference with the lambda expression
fun (x: T) => E is that the body E of the fixpoint operator may refer to itself using the name f. However,
to ensure that the function is well-defined, E must perform structural recursion on the argument x, which
means that T must be a datatype and each recursive call must descend along the structure of x. More

specifically, at each recursive call, the argument must be a value that was used to construct x. (Coq checks
this by checking that the argument is a variable bound by a match expression on x.) Since each datatype
value was constructed through a finite number of constructor calls, it follows that there can only be a finite
number of nested recursive calls, and therefore the function always terminates and has a well-defined value.

Therefore, we can define the string concatenation function as follows:

Definition concat: string -> string -> string :=
fix f(s1: string) {struct s1}: string -> string :=
fun (s2: string) =>
match s1 with

EmptyString => s2
| NonemptyString b s => NonemptyString b (f s s2)
end.

Note that the expression f s s2 should be read as (f s) s2: the result of calling function f with argument
s is a function, which is then applied to argument s2. Similarly, the type string -> string -> string
of concat should be read as string -> (string -> string): calling concat with an argument of type
string returns a function that, when itself called with an argument of type string, returns a value of type
string.

Coq accepts the fixpoint expression in the body of concat since it performs structural recursion: the
argument s of the recursive call f s was the second argument of the constructor call used to construct s1.

As a convenience, the fixpoint operator actually accepts multiple parameters; the {struct x} annotation
indicates the parameter that is reduced at each recursive call. Therefore, we can write the concatenation
function more concisely as follows:

Definition concat: string -> string -> string :=
fix f(s1 s2: string) {struct s1}: string :=
match s1 with

EmptyString => s2
| NonemptyString b s => NonemptyString b (f s s2)
end.

Coq offers syntactic sugar for a definition whose body is a fixpoint expression: the Fixpoint command.
Using this command, the definition of concat reduces to the following:

Fixpoint concat(s1 s2: string) {struct s1}: string :=
match s1 with

EmptyString => s2
| NonemptyString b s => NonemptyString b (concat s s2)
end.

5 Generics

A string is a sequence of bytes. The concat function concatenates two strings, but it doesn’t actually
depend on the fact that it is dealing with bytes; it would work just as well if it was dealing with a sequence
of booleans, a sequence of strings, etc. Ideally, the concatenation function would be generic in the type of
the elements of the sequence.

Coq supports this. We can define a generic type of sequences of elements of some type A as follows:

Inductive list(A: Type) := nil | cons(head: A)(tail: list A).

Definition hi_as_list: list byte := cons byte h (cons byte i (nil byte)).

The first command defines a datatype list which is parameterized by a type parameter A, and which has two
constructors: nil constructs an empty list, and cons constructs a non-empty list starting with element head
of type A, and whose remaining elements are given by parameter tail, which is itself a list with elements of
type A.

Note: it is not strictly correct to say that list is a datatype; rather, list is a function that takes a
type and returns a datatype. When this nuance is important, we will refer to the name being defined by an
Inductive command as a datatype constructor. This is not to be confused with the data value constructors
(or constructors for short) of a datatype (or, strictly speaking, of a datatype constructor). For example,
nil and cons are the data value constructors of the datatype constructor list. The former construct data
values (e.g. nil bool and cons bool true (nil bool)); the latter constructs datatypes (e.g. list bool
and list byte).

Note that the constructors of a generic type are automatically generic themselves: nil and cons both
take a value for the type parameter A as their first argument. This is illustrated by the definition of constant
hi_as_list, which is bound to the representation of the text “hi” as a list of bytes instead of a value of type
string. Its type is therefore list byte. Notice that both calls of constructor cons as well as the call of
constructor nil take the type byte as their first argument.

We can now define the generic version of function concat, which we shall call append:

Fixpoint append(A: Type)(l1 l2: list A) {struct l1}: list A :=
match l1 with

nil => l2
| cons h t => cons A h (append A t l2)
end.

Notice that in a match expression, only the constructor arguments for explicitly declared constructor param-
eters are bound to variables. (It would not make much sense to also bind the type argument, since it is fixed
by the type of the value being matched.)

We can check that the function works correctly by testing it:

Compute append bool (cons bool true (nil bool)) (cons bool false (nil bool)).
(* Output: = cons bool true (cons bool false (nil bool)): list bool *)

Indeed, concatenating the singleton list “true” and the singleton list “false” yields the list “true, false”.
Notice that the declaration of type parameter A of append looks just like an ordinary parameter decla-

ration, and indeed it is: Coq considers types to be values of a special type Type. So, we lied above when
we said there were two kinds of values (data values and functions) and two kinds of types (datatypes and
function types): there is a third kind of values (types) and a third kind of types (types of types, also known
as sorts). For now, it is sufficient to imagine that there is only the sort Type.

Is there also a separate type of types of types? No: since a sort is a type, its type is itself a sort: the
type of Type is simply Type.2

Notice also that there is something remarkable going on in the parameter list of append: the type of
parameters l1 and l2 depends on the value of parameter A. Recall that the type of a function that takes
an argument of type T and returns a value of type U can be denoted using arrow notation as T -> U. This
notation is not sufficient to denote the type of append; for that, we need the general dependent function
type notation forall (x: T), U, which denotes the type of functions that take an argument of type T and
return a result of type U, where the result type U may refer to the argument value by the name x. The arrow
notation T -> U is simply syntactic sugar for forall (x: T), U for the case where U does not mention x.
Therefore, we can rephrase the definition of append as follows:

Definition append: forall (A: Type), list A -> list A -> list A :=
fun (A: Type) =>
fix f(l1 l2: list A) {struct l1}: list A :=
match l1 with

nil => l2
| cons h t => cons A h (append A t l2)
end.

That is, append is a function that, when given an argument A of type Type, returns a value of type
list A -> list A -> list A.

2This is not entirely true. There is really an infinite family of sorts Type(0), Type(1), . . . and the type of Type(k) is Type(k+1).
However, Coq infers these indices behind the scenes, so in most situations one can imagine that there is just the sort Type and
that the type of Type is Type.

6 Datatypes with variable parameters

Suppose that we want to perform vector computations for a 3D game. We will want to have a type of
3D vectors. However, many operations on vectors can easily be made generic in the number of dimensions.
Ideally, we would have a single library that offered vectors of any size. Still, we want to be able to distinguish
between vectors of different dimensions in the types of our functions.

In Coq, this can be achieved. First, we define the type nat of natural numbers. A natural number is
either zero or the successor of another natural number.

Inductive nat := zero | succ(n: nat).

Definition one: nat := succ zero.
Definition two: nat := succ one.
Definition three: nat := succ two.

Then, we define the type vector, as follows:

Inductive vector(A: Type): nat -> Type :=
empty_vector: vector A zero

| nonempty_vector: forall (n: nat), A -> vector A n -> vector A (succ n).

Definition vector2d(A: Type): Type := vector A two.
Definition vector3d(A: Type): Type := vector A three.

The definition of vector is reminiscent of the definition of the generic type list: it has a type parameter A
denoting the element type, and it has two constructors: one for the empty case, and one for the nonempty
case. However, there are some differences. First of all, the result type (the type specified after the colon)
of the datatype constructor and the result types of the data value constructors are specified explicitly. In
earlier examples of datatype definitions, we omitted these. The result type of a datatype constructor defaults
to Type. The result type of a data value constructor defaults to the datatype being defined. For example,
writing these types explicitly for datatype nat would yield the following definition:

Inductive nat: Type := zero: nat | succ(n: nat): nat.

or, equivalently:

Inductive nat: Type := zero: nat | succ: nat -> nat.

The second novelty in the definition of datatype vector is that the result type of vector is not Type but
nat -> Type. In general, the result type of a datatype constructor must be of the form A1 -> ... -> An ->
Type. This form of the definition declares n variable parameters of the datatype, of types A1, ..., An. In
contrast, the parameters declared using (x: A) notation are called the fixed parameters. Type vector has
one fixed parameter, A, of type Type, and one variable parameter, unnamed, of type nat. As a result, to
construct a datatype, vector must be applied to a type and a value of type nat.

The difference between a fixed datatype parameter and a variable datatype parameter is as follows: in the
types of the data values constructed by the data value constructors, the value for the fixed parameters is the
same for all constructors. For example, in the definition of vector, both empty_vector and nonempty_vector
construct data values of type vector A .... To use another value for the fixed parameter would be an
error. In contrast, the values for the variable parameters may differ among constructors. For example,
empty_vector constructs data values of type vector A zero whereas nonempty_vector constructs data
values of type vector A (succ n) for some n.

Notice furthermore that the type of nonempty_vector is a dependent type: the constructor takes three
arguments, and the type of the third argument, as well as the return type, depend on the first argument n.

We can now define our vector computation library, with functions that are polymorphic in the size of the
vector, but whose type nonetheless specifies the incoming and outgoing vector sizes. Function negate_vector
below illustrates this:

Fixpoint negate_vector(n: nat)(v: vector bool n) {struct v}: vector bool n :=
match v in vector _ N return vector bool N with

empty_vector => empty_vector bool
| nonempty_vector m b w => nonempty_vector bool m (negate b) (negate_vector m w)
end.

Compute
negate_vector two

(nonempty_vector bool one false
(nonempty_vector bool zero true

(empty_vector bool))).
(* Output:

= nonempty_vector bool one true
(nonempty_vector bool zero false

(empty_vector bool))
: vector bool two *)

The idea of function negate_vector is simple: for a given n, negate_vector n takes a vector of booleans
of size n and returns a copy of that vector with each element negated. The implementation, too, is a
straightforward recursive one. The only remarkable elements of this function are the in and return clauses
of the match expression. These aid Coq in checking the well-typedness of the match expression. In the
simple case, when type-checking a match expression, Coq type-checks each case and then checks that all cases
return the same type; finally, it then uses this type as the type of the match expression itself. For the match
expression in negate_vector, this does not work: the return type of the first case is vector bool zero;
the return type of the second case is vector bool (succ m) for some m. These are different types, so the
straightforward check would fail.

To type-check the match expression in negate_vector, Coq should take into account the fact that the
value being matched against is of type vector bool n, and that if this value is empty_vector bool, then
n must equal zero, and that if this value is nonempty_vector bool m b w, then n must equal succ m, and
that therefore the return type of both cases equals vector bool n. In other words, the match expression’s
return type depends on the value for the variable parameter in the type of the value being matched against.
The in and return clauses of the match expression make this dependency explicit, enabling Coq to type-
check the expression. Specifically, the in clause assigns names to the values for the variable parameters in
the type of the value being matched against; the return clause can use these names to specify the return
type of the match expression. When type-checking each case, the case’s return expression is type-checked
against the specified return type, after substituting the variable arguments in the constructor’s type for the
placeholder names. The type of the entire match expression is the specified return type, after substituting
the variable arguments in the type of the expression being matched against for the placeholder names.

7 Types are propositions

In Coq, not all types are inhabited, i.e. for some types, there is no value of that type. For example, it is
perfectly legal to declare an inductive datatype with zero constructors:

Inductive empty_type :=.

There is no value of this type; it is uninhabited.
The inhabitedness or uninhabitedness of a given type is not always immediately obvious. We have the

following properties:

• A datatype is inhabited iff3 it has at least one constructor whose parameter types are inhabited (taking
into account any dependencies between the parameters).

• A non-dependent function type is inhabited iff the parameter type is uninhabited or the return type is
inhabited. (A function type whose parameter type is uninhabited has a value: the empty function.)

3if and only if

• A dependent function type is inhabited iff, for each value of the parameter type, the corresponding
return type is inhabited.

Notice that these properties of the inhabitedness of Coq types are very similar to the properties of the
truth of logical statements (or propositions). Indeed, the non-dependent function types correspond exactly to
the logical implications: an implication P -> Q is true iff P is false or Q is true. Furthermore, the dependent
function types correspond exactly to universal quantification: forall (x: T), U is true iff for each value x
of type T, U is true.

We can easily encode the other logical connectives as datatype constructors:

Inductive And(P Q: Type) := and_(p: P)(q: Q).
Inductive Or(P Q: Type) := or_left(p: P) | or_right(q: Q).
Inductive Exists(T: Type)(P: T -> Type) := exists_(t: T)(p: P t).
Inductive Equals(A: Type)(a: A): A -> Type := equals_: Equals A a a.
Inductive False :=.
Inductive True := true_.

Indeed, type And P Q is inhabited iff P is inhabited and Q is inhabited; type Or P Q is inhabited iff P is
inhabited or Q is inhabited; Exists T P is inhabited iff there exists a value t of type T such that P t is
inhabited; Equals A a b is inhabited iff a equals b; False is uninhabited; True is inhabited.

We can encode negation as follows:

Definition Not(P: Type) := P -> False.

Indeed, Not P is inhabited iff P is uninhabited.
Using this trick (known as the Curry-Howard correspondence) we can encode some properties of the

functions that we have defined above as Coq types:

forall (b: bool), Equals bool (negate (negate b)) b
forall (s: string), Equals string (concat s EmptyString) s

8 Expressions are proofs

If inhabitedness of a type in Coq corresponds to truth of a proposition in logic, then what is the Coq
equivalent of a logical proof? Proving the truth of a proposition corresponds to proving the inhabitedness
of a type. The latter can be done by showing a witness, i.e. a value of the type. In fact, it is sufficient to
show a Coq expression of the type: since evaluation of Coq expressions always terminates and never crashes,
every expression of some type evaluates to a value of that type.

Let’s see what this means for the various logical connectives discussed above. For an implication P -> Q,
a proof is a function that takes an argument of type P (i.e. a proof of proposition P) and computes a value
of type Q; in other words, in order to prove P -> Q, it is sufficient to prove Q assuming P. Similarly, for
a universal quantification forall (x: T), P, a proof is a function that, for any given value x of type T,
computes a proof of P (where x may appear in P).

Conversely, an implication P -> Q that is available as an assumption in a proof can be used to prove the
goal by applying its proof to a proof of P, thus obtaining a proof of Q. Similarly, when given a proof of a
universal quantification forall (x: T), P, applying the proof to a value of type T yields a proof of P with
that value substituted for x.

As a concrete example, let’s prove the proposition forall (P Q: Type), P -> (P -> Q) -> Q. In
words: for any two propositions P and Q, if we have P and we have that P implies Q, then we have Q. The
proof is very simple:

Definition modus_ponens: forall (P Q: Type), P -> (P -> Q) -> Q :=
fun (P Q: Type) =>
fun (p: P) =>
fun (pq: P -> Q) =>
pq p.

The proof computes a value of type Q by applying the given function of type P -> Q to the given value of
type P, thus proving the inhabitedness of Q.

To prove a conjunction And P Q, one applies the constructor and_ to a proof of P and a proof of Q;
similarly, to prove a disjunction Or P Q, one either applies the constructor or_left to a proof of P, or one
applies the constructor or_right to a proof of Q. To prove an existential quantification Exists T P, one
applies the constructor exists_ to a witness t and a proof of P t.

Conversely, when given a proof of a conjunction And P Q, one can obtain a proof of P and a proof of Q
using a match expression on the proof of And P Q; similarly, when given a proof of a disjunction Or P Q, and
when proving a goal G, one can obtain a proof of G in the form of a match expression on the proof of Or P Q
that in the first branch proves G from P and in the second branch proves G from Q. When given a proof of
an existential quantification Exists T P, one can extract a witness t and a proof of P t through a match
expression.

We illustrate this with a proof of the distributivity of conjunction over disjunction:

Definition distr: forall (P Q R: Type), And P (Or Q R) -> Or (And P Q) (And P R) :=
fun (P Q R: Type) =>
fun (pqr: And P (Or Q R)) =>
match pqr with
and_ p qr =>
match qr with

or_left q =>
or_left (And P Q) (And P R) (and_ P Q p q)

| or_right r =>
or_right (And P Q) (And P R) (and_ P R p r)

end
end.

To prove an equality Equals A a b, one simply applies the constructor equals_ A a, provided that Coq
can reduce a and b to the same expression. To illustrate this, consider the property

forall (b: bool), Equals bool (negate (negate b)) b

The following proof attempt fails:

Definition negate_negate: forall (b: bool), Equals bool (negate (negate b)) b :=
fun (b: bool) =>
equals_ bool b.

(* Fails *)

This is because Coq cannot reduce negate (negate b); to reduce the match expression in the body of
negate, Coq needs to know the value of b. Therefore, we need to perform case analysis on b, using a match
expression:

Definition negate_negate: forall (b: bool), Equals bool (negate (negate b)) b :=
fun (b: bool) =>
match b as B return Equals bool (negate (negate B)) B with

true => equals_ bool true
| false => equals_ bool false
end.

This definition is accepted by Coq, i.e. this proof succeeds. By using the combination of an as clause and a
return clause, we can substitute true resp. false for b in the return type (i.e. the proof goal) for the purpose
of type-checking the first, resp. the second branch of the match expression. Since negate (negate true)
reduces to true, Coq considers the types Equals bool (negate (negate true)) true and Equals bool
true true to be equal.

When given a proof of an equality, one can use it to rewrite the goal. We prove symmetry of equality:

Definition symm(A: Type)(x y: A)(xy: Equals A x y): Equals A y x :=
match xy in Equals _ _ Y return Equals A Y x with

equals_ => equals_ A x
end.

Thanks to the in and return clauses introduced in Section 6, inside the match expression the goal becomes
Equals A x x, which is proved directly using the constructor.

9 Inductive proofs

Let us now try to prove that concatenating an empty string to the end of a string does nothing:

forall (s: string), Equals string (concat s EmptyString) s

We will need to perform case analysis on s so that Coq can reduce the concat call, which performs a match
expression on its first argument. We get two branches. The EmptyString branch is dealt with trivially. In
the NonemptyString branch, the goal becomes

Equals string (concat (NonemptyString b s’) EmptyString) (NonemptyString b s’)

This reduces to

Equals string (NonemptyString b (concat s’ EmptyString)) (NonemptyString b s’)

At this point, we can finish the proof if only we can use the fact that concat s’ EmptyString equals s’.
But this is exactly what we are proving. Fortunately, s’ was used as an argument in the constructor call used
to construct s; therefore, we can use the fixpoint operator introduced in the section on recursive datatypes:

Definition concat_EmptyString:
forall (s: string), Equals string (concat s EmptyString) s :=

fix f(s: string) {struct s}: Equals string (concat s EmptyString) s :=
match s as S return Equals string (concat S EmptyString) S with

EmptyString => equals_ string EmptyString
| NonemptyString b s’ =>

match f s’ in Equals _ _ S’
return Equals string (NonemptyString b (concat s’ EmptyString)) (NonemptyString b S’)
with

equals_ => equals_ string (NonemptyString b (concat s’ EmptyString))
end

end.

Notice that this proof has the structure of an inductive proof: there is a base case and an inductive step. In
the inductive step, we have access to an induction hypothesis, in the form of the recursive call.

10 Inductively defined predicates

A concept that is pervasive in formal systems is the notion of an inductively defined predicate, also known as
an inductively defined judgment/relation/set. In conventional terminology, such a predicate is defined by a
set of axioms and inference rules, specifying the ways in which instances of the predicate can be constructed
or derived. Inference rules differ from axioms in that the latter have premises and the former do not. Premises
may themselves be instances of the predicate or not. (Often, premises that are not instances of the predicate
themselves are called side conditions and rules that have side conditions only are considered to be axioms.)

For example, the less-than-or-equal-to relation on natural numbers can be defined inductively as follows
(in the conventional inference rule notation):

Le-Refl

n ≤ n

Le-Succ
m ≤ n

m ≤ n+ 1

Note: the free variables in an axiom or inference rule are implicitly universally quantified. In other words,
there is a separate instance of the rule for each instantiation of the free variables of a rule.

Given such a definition, a particular argument list satisfies the predicate if and only if a derivation tree (or
proof tree) can be built whose root is the predicate applied to this argument list. A derivation tree is a finite
tree where each node is labeled by an instance of the predicate which 1) matches an instance of an axiom
or 2) matches the conclusion of an instance of an inference rule, and where for each (non-side-condition)
premise of the instance of the inference rule, the node has a child whose label matches the premise.

For example, here is a derivation tree for the proposition 0 ≤ 2:

Le-Refl
0 ≤ 0

Le-Succ
0 ≤ 1

Le-Succ
0 ≤ 2

The conventional concept of an inductively defined predicate corresponds exactly to a Coq parameterized
datatype interpreted as a proposition via the Curry-Howard correspondence. Each axiom and each inference
rule corresponds to a constructor of the datatype.

For example, the definition of the less-than-or-equal-to relation is as follows:

Inductive Le: nat -> nat -> Type :=
le_refl: forall (n: nat), Le n n

| le_succ: forall (m n: nat), Le m n -> Le m (succ n).

The free variables and the premises of a rule correspond to the parameters of the constructor.
A derivation tree corresponds exactly to a value built from the constructors of the datatype. For example,

here is a proof of Le zero two:

Definition Le_zero_two: Le zero two :=
le_succ zero one

(le_succ zero zero
(le_refl zero)).

To prove a property of an inductively defined predicate, one typically performs induction on the size of
the derivation tree (or, for short, induction on the derivation). This corresponds in Coq with structural
recursion on the proof of the predicate instance using the fixpoint operator that we saw before. For example,
here is a proof of the property m ≤ n⇒ m+ 1 ≤ n+ 1:

Fixpoint Le_succ_succ(m n: nat)(H: Le m n) {struct H}: Le (succ m) (succ n) :=
match H in Le M N return Le (succ M) (succ N) with

le_refl n0 =>
(* Goal: Le (succ n0) (succ n0) *)
le_refl (succ n0)

| le_succ m0 n0 H0 =>
(* Goal: Le (succ m0) (succ (succ n0)) *)
le_succ (succ m0) (succ n0) (Le_succ_succ m0 n0 H0)

end.

11 Coq versus classical logic

Coq’s built-in logic, whose essence was presented above, lacks some properties that are occasionally very
useful. The main one is undoubtedly the law of the excluded middle: forall (P: Type), Or P (Not P).
In Coq terms, this proposition states that for any type, either it is inhabited or a function exists from that
type to False, a manifestly uninhabited type. This and other propositions that are true in classical logic,
and consistent with Coq’s own logic, but not provided by it, can be declared as axioms:

Axiom excluded_middle: forall (P: Type), Or P (Not P).

This axiom enables a very important proof technique: proof by contradiction.4

Another useful axiom is functional extensionality, the principle that if two functions are pointwise equal,
then they are equal:5

Axiom fun_ext:
forall (A B: Type)(f g: A -> B),
(forall (x: A), Equals B (f x) (g x)) ->
Equals (A -> B) f g.

A third useful axiom is the axiom of choice, stating that if a relation is total (i.e. it relates each value to
some value) then it defines a function.6

Axiom choice:
forall (A B: Type)(R: A -> B -> Type),
(forall (x: A), Exists B (R x)) ->
Exists (A -> B) (fun (f: A -> B) => forall (x: A), R x (f x)).

12 Limitations of the Curry-Howard correspondence

It should be clear from the preceding sections that the Curry-Howard correspondence is very powerful: a
proof system for a very large subset of classical logic can be obtained by interpreting programming language
types as propositions and programs as proofs. However, the correspondence has its limitations: not all of
classical logic can be obtained this way. In the previous section, we saw some properties of classical logic
that are not provided by the correspondence, but that can be introduced as axioms. However, there are
other properties of classical logic that are inconsistent with the correspondence.

12.1 Propositions about propositions

We encounter one such inconsistency when we consider propositions about propositions, i.e. propositions
that quantify over propositions. For example, consider the proposition that states that for any proposition
P , P implies P ∧ P . In Coq, we can state and prove this proposition as follows:

Definition P0: Type := forall (P: Type), P -> And P P.

Definition P0proof: P0 :=
fun (P: Type) =>
fun (p: P) =>
and_ P P p p.

So far, so good. However, we should now be able to apply this theorem to itself to obtain the property

(∀P. P → P ∧ P ) ∧ (∀P. P → P ∧ P )

as follows:

Definition AndP0P0proof: And P0 P0 := P0proof P0 P0proof.
(* Error: Universe inconsistency. *)

Unfortunately, Coq refuses to accept this definition; it complains that this definition would create a universe
inconsistency.

To understand this error message, recall that our statement that the type of a type is Type was a little
white lie: in fact, there is an infinite hierarchy of sorts Type(0), Type(1), Type(2), etc., and the type of a
given type is as follows:

4This axiom appears in module Classical_Prop in the Coq standard library with P: Prop instead of P: Type. It seems fine
to use P: Type here, but use at your own risk.

5This axiom appears in module FunctionalExtensionality.
6This axiom appears in module ClassicalEpsilon with R: A -> B -> Prop.

• The type of a datatype is the maximum of the types of the parameter types of the constructors (not
including the fixed datatype parameters).

• The type of a function type is the maximum of the type of the parameter type and the type of the
result type.

• The type of a sort Type(k) is Type(k+1).

Coq does not require or allow us to write these indices ourselves; it infers them automatically such that
the abovementioned universe constraints are satisfied. If it cannot, it reports a universe inconsistency.

In the example, let k be the index of the type of parameter P in P0. Then the type of And P P is
Type(k), and the type of P0 is Type(k+1). As a result, since P0proof takes only values of type Type(k) as
an argument, it cannot accept P0.

Why does Coq impose these constraints? What is wrong with a system where there is one sort Type
that is the type of all types? The answer is that such a system is inconsistent: it allows expressions to be
constructed of any type whatsoever, including False.

Indeed, assume that there is one sort Type which is the type of all types. Then we can prove False as
follows.7 We will encode naive set theory and then exploit Russell’s paradox to obtain False. First, we
define the universe U of all sets:

Inductive U := set(I: Type)(P: I -> Type)(f: I -> U).

Definition In(x X: U): Type :=
match X with
set I P f => Exists I (fun i => And (P i) (Equals U (f i) x))

end.

The proposition In x X encodes x ∈ X. Note that the above definitions are perfectly valid in Coq; they do
not violate any universe constraints.

We can now define sets by specification:

Definition id(x: U): U := x.
Definition spec(P: U -> Type): U := set U P id.

Definition spec_intro(P: U -> Type)(x: U)(H: P x): In x (spec P) :=
exists_ U (fun (i: U) => And (P i) (Equals U (id i) x))

x (and_ (P x) (Equals U (id x) x) H (equals_ U (id x) x)).
Definition spec_elim(P: U -> Type)(x: U)(H: In x (spec P)): P x :=

match H with exists_ i H0 =>
match H0 with and_ H1 H2 =>
match H2 in Equals _ _ X return P X with equals_ =>
H1
end end end.

We proved that In x (spec P) if and only P x, for all sets x. Note that the definition of spec would trigger
a universe inconsistency error in Coq.

We can now easily derive False by considering the set R of all sets that do not contain themselves:

Definition PR(x: U): Type := Not (In x x).
Definition R: U := spec PR.

Definition contradiction(P: Type)(H1: Not P -> P)(H2: P -> Not P): False :=
(fun (H3: Not P) => H3 (H1 H3)) (fun (HP: P) => H2 HP HP).

Definition absurd: False := contradiction (In R R) (spec_intro PR R) (spec_elim PR R).

The abovementioned universe constraints enable an interpretation of a type as the set of its values and a
function as the set of its argument-result pairs in set theory: the interpretation of a type can be defined by

7Based on a proof by Paolo Capriotti.

recursing on the universe index and then on the structure of the type. Allowing Type : Type would preclude
such an interpretation since this would mean that the interpretation of type Type contains itself, which is
not possible in standard set theory.

12.2 Proof irrelevance

Another such inconsistency appears when we want to define subset types, i.e. types consisting of values that
satisfy some property.

For example, we could define the integers as the pairs of natural numbers where at least one is zero:

Inductive Z :=
z_(a b: nat)(H: Or (Equals nat a zero) (Equals nat b zero)).

Positive and negative 1 have the following representation in Z:

Definition pos1: Z :=
z_ one zero

(or_right (Equals nat one zero) (Equals nat zero zero) (equals_ nat zero)).
Definition neg1: Z :=

z_ zero one
(or_left (Equals nat zero zero) (Equals nat one zero) (equals_ nat zero)).

Is Z isomorphic with Z? Unfortunately, no: there are two distinct representations of the number zero:

Definition zero1: Z :=
z_ zero zero

(or_left (Equals nat zero zero) (Equals nat zero zero) (equals_ nat zero)).
Definition zero2: Z :=

z_ zero zero
(or_right (Equals nat zero zero) (Equals nat zero zero) (equals_ nat zero)).

Notice that zero1 and zero2 differ only in their third argument: the proof that at least one of the constituent
natural numbers is zero. To show that these two values are indeed considered to be distinct by Coq, we can
define a function that distinguishes them:

Definition is_zero1(z: Z): bool :=
match z with
z_ zero zero (or_left _) => true

| _ => false
end.

Definition zero1_is_zero1: Equals bool (is_zero1 zero1) true :=
equals_ bool true.

Definition zero2_is_not_zero1: Equals bool (is_zero1 zero2) false :=
equals_ bool false.

In order to facilitate the definition of subset types like Z in which appropriate equalities hold, we would like
to have proof irrelevance: two proofs of the same proposition are considered equal. It would be tempting to
introduce this property as an axiom; however, without some means to distinguish types used as propositions
from types whose values have information content, such an axiom would be inconsistent.

12.3 Propositional extensionality

In classical logic, two propositions that are both true or both false can be considered equal. For example, in
classical logic we have True∨True = True. However, in Coq we have Not (Equals Type (Or True True) True):
Or True True is a type of two elements (or_left True True true_ and or_right True True true_)
whereas True is a type of one element (true_). The proof follows:

Definition singleton(A: Type): Type := forall (a b: A), Equals A a b.

Definition True_singleton: singleton True :=
fun (p q: True) =>
match p as P return Equals True P q with

true_ =>
match q as Q return Equals True true_ Q with

true_ =>
equals_ True true_

end
end.

Definition Not_Or_True_True_singleton: Not (singleton (Or True True)) :=
fun (p: singleton (Or True True)) =>
match

match p (or_left True True true_) (or_right True True true_) in Equals _ _ B
return match B with or_left _ => True | or_right _ => False end
with

equals_ => true_
end

return False with end.

Definition Not_Equals_Or_True_True_True: Not (Equals Type (Or True True) True) :=
fun (p: Equals Type (Or True True) True) =>
match p in Equals _ _ B return Not (singleton B) with

equals_ => Not_Or_True_True_singleton
end

True_singleton.

13 Beyond Curry-Howard: sort Prop

To address the limitations identified in the preceding section, Coq provides a means to allow the user
to distinguish between types whose values have informational value on the one hand, and types used as
propositions on the other hand: the latter types should be declared to be of sort Prop. For example:

Inductive AndP(P Q: Prop): Prop := andp_(p: P)(q: Q).
Inductive OrP(P Q: Prop): Prop := orp_left(p: P) | orp_right(q: Q).
Inductive EqualsP(A: Type)(a: A): A -> Prop := equalsp_: EqualsP A a a.
Inductive TrueP: Prop := truep_.
Inductive FalseP: Prop :=.

13.1 The elimination restriction

Coq enforces that values of types of sort Prop are not used for their informational content by imposing an
elimination restriction: if a match expression matches on a value of a type of sort Prop, the return type of
the match expression must itself also be of sort Prop. In other words, matching on proofs is allowed only to
build other proofs.8 For example:

Definition is_orp_left(P Q: Prop)(H: OrP P Q): Prop :=
match H with
orp_left _ => TrueP

| orp_right _ => FalseP
end.

8An exception is if the type being matched on has just a single constructor and this constructor’s parameter types are
themselves of sort Prop. Such types have no informational content anyway.

15



(*
Error:
Incorrect elimination of "H" in the inductive type "OrP":
the return type has sort "Type" while it should be "Prop".
Elimination of an inductive object of sort Prop
is not allowed on a predicate in sort Type
because proofs can be eliminated only to build proofs.

*)

Thanks to this restriction, a type of sort Prop can be interpreted in set theory as either the empty set (if the
type is uninhabited) or a singleton set containing a canonical universal proof object (if the type is inhabited).
The value of a match expression that matches on a proof is always simply this universal proof object.

As a result, the limitations identified in the preceding section do not hold for sort Prop.

13.2 Benefits

It is possible to apply propositions about propositions to themselves:

Definition P0P: Prop := forall (P: Prop), P -> And P P.

Definition P0Pproof: P0P :=
fun (P: Prop) =>
fun (p: P) =>
and_ P P p p.

Definition AndPP0PP0Pproof: AndP P0P P0P := P0Pproof P0P P0Pproof.

This typechecks because if the result type of a function type is of sort Prop, then the function type itself is
of sort Prop.

We can define subset types with appropriate equalities9:

Inductive ZP :=
zp_(a b: nat)(H: OrP (EqualsP nat a zero) (EqualsP nat b zero)).

Definition one := succ zero.
Definition pos1: ZP :=

zp_ one zero
(orp_right (EqualsP nat one zero) (EqualsP nat zero zero) (equalsp_ nat zero)).

Definition neg1: ZP :=
zp_ zero one

(orp_left (EqualsP nat zero zero) (EqualsP nat one zero) (equalsp_ nat zero)).
Definition zero1: ZP :=

zp_ zero zero
(orp_left (EqualsP nat zero zero) (EqualsP nat zero zero) (equalsp_ nat zero)).

Definition zero2: ZP :=
zp_ zero zero

(orp_right (EqualsP nat zero zero) (EqualsP nat zero zero) (equalsp_ nat zero)).

Axiom proof_irrelevance: forall (P: Prop)(p q: P), EqualsP P p q.

Definition EqualsP_zero1_zero2: EqualsP ZP zero1 zero2 :=
match

proof_irrelevance
(OrP (EqualsP nat zero zero) (EqualsP nat zero zero))
(orp_left (EqualsP nat zero zero) (EqualsP nat zero zero) (equalsp_ nat zero))
(orp_right (EqualsP nat zero zero) (EqualsP nat zero zero) (equalsp_ nat zero))

9This axiom is available in the Coq standard library in module ProofIrrelevance.

16



in EqualsP _ _ Q return EqualsP ZP zero1 (zp_ zero zero Q)
with

equalsp_ => equalsp_ ZP zero1
end.

We can perform equational reasoning on propositions:

Axiom propext: forall (P Q: Prop), (P -> Q) -> (Q -> P) -> EqualsP P Q.

13.3 Relaxing the elimination restriction

The Prop elimination restriction generally does not cause difficulties in practice. There are, however, a few
exceptions. One is that we sometimes want to define a function by case splitting on the truth or falsehood
of a proposition. For example, here is an attempt to define function update:

f [x := y] = λa.

{
y if a = x
f(a) otherwise

Axiom excluded_middle: forall (P: Prop), OrP P (NotP P).

Definition update(A B: Type)(f: A -> B)(x: A)(y: B): A -> B :=
fun (a: A) =>
match excluded_middle (EqualsP A a x) with

orp_left _ => y
| orp_right _ => f x
end.

(* Error: Incorrect elimination of "excluded_middle (EqualsP A a x)" ... *)

There are two possible solutions. One solution is to develop a decision procedure for the proposition (i.e. an
expression of type bool instead of type Prop). Another is to use a version of the axiom of excluded middle
where the disjunction is in Type rather than in Prop10:

Axiom excluded_middle_informative: forall (P: Prop), Or P (NotP
P).

Definition update(A B: Type)(f: A -> B)(x: A)(y: B): A -> B :=
fun (a: A) =>
match excluded_middle_informative (EqualsP A a x) with
or_left _ => y

| or_right _ => f x
end.

Definition update_test: EqualsP nat (update nat nat succ zero zero zero) zero :=
match excluded_middle_informative (EqualsP nat zero zero) as E
return

EqualsP nat
(match E with or_left _ => zero | or_right _ => succ zero end) zero

with
or_left _ =>
equalsp_ nat zero

| or_right H =>
match H (equalsp_ nat zero) return EqualsP nat (succ zero) zero with end

end.

10This axiom is available in module ClassicalEpsilon in the Coq standard library.

Another place where one bumps into the elimination restriction is when one wishes to define a function by
indefinite description. For example:

Definition get_witness(T: Type)(P: T -> Prop)(e: ExistsP T P): T :=
match e with
existsp_ t p => t

end.
(* Error: Incorrect elimination of "e" ... *)

Again, there are two possible solutions. One solution is to develop an algorithm that computes a witness.
Another is to use an axiom that converts an existence proof in Prop to one in Type11:

Axiom constructive_indefinite_description:
forall (T: Type)(P: T -> Prop), ExistsP T P -> Exists T P.

Definition get_witness(T: Type)(P: T -> Prop)(e: ExistsP T P): T :=
match constructive_indefinite_description T P e with

exists_ t p => t
end.

13.4 Subtyping

Notice that something strange is happening in the expressions above. In the statement of axiom excluded_-
middle_informative, we pass P and NotP P as arguments to Or, even though Or’s parameters are of type
Type and P and NotP P are of type Prop. Similarly, in the statement of axiom constructive_indefinite_-
description, we are passing P as an argument to Exists. What is going on here is a form of subtyping :
any type is of type Type, even if it is also of type Prop. That is, Prop is a subtype of Type. And in fact, the
type of Prop itself is Type(1).

14 The sort Set

In addition to the hierarchy of Type sorts and the sort Prop, Coq has the sort Set. As far as this author
can see, Set is simply a way to write Type(0). It is therefore not really a distinct sort. The type of Set is
Type(1). From the fact that Set is simply Type(0), it follows that Set is the type of the “small” datatypes
and function types, i.e. the ones whose values do not directly or indirectly involve types.

It seems to be the case that for any Coq file that mentions Set and that verifies, if one replaces all
occurrences of Set with Type the file still verifies. Therefore, the notation Set seems to be superfluous.
Reasons why it exists nonetheless may be the following: 1) historical/compatibility: perhaps old versions of
Coq had Set but not Type; 2) there is a command line option that makes Set impredicative, but not Type;
3) it may have (probably very minor) performance and simplicity benefits since using Set instead of Type
avoids the introduction of universe level variables and constraints.

Note: It does not seem to be the case that expressions of types of sort Set are treated differently
from expressions of types of sort Type by the extraction mechanism (that translates Coq definitions to the
programming language OCaml). Indeed, many expressions that should be, and are, extracted successfully
are of types of sort Type(k) with k > 0, e.g. type-generic functions such as list concatenation.

15 Conclusion

You are now in a position where you can see how a wide variety of theory developments (definitions and
proofs) can in principle be formalised in Coq. The main intended results to take away from reading this text
are an understanding and an appreciation of precise, rigorous reasoning: simple rules can be given for the
form and validity of an argument or line of reasoning.

11This axiom, too, is available in module ClassicalEpsilon in the Coq standard library.

This text did not introduce the many features that Coq offers to facilitate the definition of concepts
(type inference, implicit arguments, notations, sections, modules, etc.) or the interactive construction of
proofs (proof scripts, the tactic language, automation, reflection, etc.) or the extraction of programs into
an industrial-strength programming language like OCaml. It also did not introduce the many concepts
available in the Coq standard library. Finally, it did not present all details of Coq’s logic, the Calculus of
Inductive Constructions; missing elements include mutually recursive datatypes and functions, and coinduc-
tive datatypes and functions. For more information on all of these, see the Coq website, or the books by
Yves Bertot, Benjamin Pierce, or Adam Chlipala.

